

Abstract

Several years ago Leon Magiera produced a very extended paper "DERIVE for Physics" treating problems from electric and magnetic fields using the at this times available and widely used CAS DERIVE. Josef Boehm translated the paper and added CAS-parts using TI-Voyage and the first versions of TI-Nspire. The German and English book were ready to be printed and published. (This was 2006 / 2007). Then DERIVE was taken off the market and the publisher ended his business … End of the story? No!

Just recently Leon sent a new paper to Josef. He does not want to leave his paper hidden in his room. He rewrote the paper based on the free CAS wxMaxima and offered Josef to set his "Maxima for Physics" anywhere in the Internet for free download. His paper from 2006 was extended and the earlier chapters about problems from Electric Fields and Magnetic Fields are now accompanied by chapters "Circuits" and " Mechanics of a Charge in Electric and Magnetic Fields".

Josef added solving the problems not only using Maxima, but also TI-NspireCAS and sometimes good old DERIVE focussing on the advantages and disadvantages of the various software tools. So he changed the title to a more general "CAS" for Physics Problems.

We will present a selection of examples from all four fields covered in the papers.

email address of the authors:

nojo.boehm@pgv.at

leon.magiera@wp.pl

This was the cover of the German version in 2006.

Derive®
für Physik 1

Elektrizität und Magnetismus

Leon Magiera

übersetzt und bearbeitet von Josef Böhm

bk teachware Schriftenreihe Nr.

The new version comprises four parts and an introduction, Electrostatics, Magnetism, Circuits and Mechanics of Charged Particles.

I – Josef – am no physicist. So my part was to cope with CAS-questions, comparisons between various computer algebra systems, pose "silly questions" and to bring the whole paper into a common format.

Introduction

- This presentation is devoted to solving practical problems in the field of electricity and magnetism using computer algebra systems (Maxima (maxima.sourceforge.net), TI-Nspire, DERIVE).
- The selected problems usually appear in standard general physics courses at university level (science and engineering); some are also suitable for high schools.
- Each section begins with the formulation of a physical problem and then the reader is lead through a detailed, step by step, description of its solution with the use of the computer algebra system.

Example 1: Electric Field

II.6 There are five charges Q_1, Q_2, Q_3, Q_4 and Q_0 located as shown in Fig. II.6.



Fig. II.6

a) Find the relation between charges Q_1, Q_2, Q_3, Q_4 for which the resulting acting on the charge Q_0 disappears?

b) Plot the equipotential curves and the 3D-plot of the potential.

One of the solutions, namely: $Q_1 = Q_2 = Q_3 = Q_4$ (full symmetry) is obvious. We try to find the remaining solutions:

```
(%i1) R_:matrix([a,a,0],[-a,a,0],[-a,-a,0],[a,-a,0])/2$
      assume(a>0)$
      r_:[x,y,z]$  Q_:[Q1,Q2,Q3,Q4]$
```

The potential of the electric field at point r resulting from the charges Q_i placed at R_i is given by

$$\Phi_j = \frac{1}{4\pi\varepsilon_0}\frac{Q_j}{\left|\vec{r}-\vec{R}_j\right|}$$

```
(%i5) Phi:1/(4*%pi*eps0)*
            sum(Q_[j]/sqrt((r_-R_[j]).(r_-R_[j])),j,1,4)$
```

We apply the relationship between potential and field strength:

```
(%i6) load(vect)$  E_:-grad(Phi)$
(%i8) E_:ev(express(E_),diff)$
      E0_:ev(E_,x=0,y=0,z=0)$
(%i10) E0x:factor(E0_[1]);E0y:factor(E0_[2]);E0z:factor(E0_[3]);
```

(%o10) $-\frac{Q4-Q3-Q2+Q1}{2^{3/2}\pi a^2 eps0}$

(%o11) $\frac{Q4+Q3-Q2-Q1}{2^{3/2}\pi a^2 eps0}$

(%o12) 0

The electric field vector disappears in the desired point if every its component is zero. This implies that we have to solve two equations. We try to solve this equation system e.g. for the unknowns Q_1 and Q_2.

```
(%i13) solve([E0x,E0y],[Q1,Q2]);
(%o13) [[Q1=Q3,Q2=Q4]]
```

From the received solution we can conclude that the resultant power acting on charge Q_0 disappears when conditions $Q_1 = Q_3$ and $Q_2 = Q_4$ are fulfilled.

Of course, if every component of any vector is equal zero then its value (length of the vector) is equal to zero. We are getting the task to the solution of only one equation

$$\left(\vec{E}\vec{E}\right)_{\vec{r}=[0,0,0]}=0\,.$$

We try to solve this equation for one unknown e.g. for Q_1:

```
(%i14) solve(E0_.E0_,Q1);  solve(E0_.E0_,Q2);
(%o14) [Q1=-%i Q4+Q3+%i Q2,Q1=%i Q4+Q3-%i Q2]
 (%o15) [Q2=Q4-%i Q3+%i Q1,Q2=Q4+%i Q3-%i Q1]
```

As charges are real we deduce from `%o14` that $Q_1 = Q_3$ and further $Q_2 = Q_4$.

Of course, the same relations between charges are being received by solving the equation with respect to any other charge e.g. Q_2 (see above `%o15`).

We can obtain the solution of the problem in a single step applying the `solve` command:

```
(%i16) solve(E0_,[Q1,Q2]);
solve: dependent equations eliminated: (3)
 (%o16) [[Q1=Q3,Q2=Q4]]
```

Now we substitute appropriate data and plot the potential:

```
(%i17) eps0:1$  a:1$  Q1:1$  Q2:-1$  Q3:1$  Q4:Q2$
(%i23) Phiz0:ev(Phi,z=0)$
(%i24) load(draw)$
```

```
(%i27) wxplot3d(Phiz0,[x,-2,2],[y,-2,2],[color,magenta]);
```

(%t27)



(%o27)



The equipotential curves in the *xy*-plane are plotted:

```
(%i26) draw3d(title="Contours of the Potential",
       explicit(Phiz0,x,-1,1,y,-1,1),
       contour_levels = 25,
       contour = map,
       surface_hide = true);
```



These are the respective DERIVE Plots:





TI-Nspire does not enable plotting the contour lines.

We introduce sliders for the *xy*-plane for visualizing the contour lines:







It is interesting that the old handheld Voyage 200 enables even plotting the contour lines – not in best quality, but it works:







I transfer the equation of the potential to GeoGebra and plot the surface together with contour lines:

```
(%i27)   Phiz0:ev(Phi,z=0);
```

$$(\text{Phiz0})\quad \frac{\dfrac{1}{\sqrt{\left(y+\frac{1}{2}\right)^2+\left(x+\frac{1}{2}\right)^2}}-\dfrac{1}{\sqrt{\left(y+\frac{1}{2}\right)^2+\left(x-\frac{1}{2}\right)^2}}-\dfrac{1}{\sqrt{\left(y-\frac{1}{2}\right)^2+\left(x+\frac{1}{2}\right)^2}}+\dfrac{1}{\sqrt{\left(y-\frac{1}{2}\right)^2+\left(x-\frac{1}{2}\right)^2}}}{4\,\pi}$$



Example 2: Electric Field

A cylindrical dielectric layer characterized by two radii R_1, R_2 and height H is uniformly charged with a charge Q. Find the vector of the electric field

a) on the axis of symmetry,

b) in the centre of the base circle (see Figure).



The resultant vector of the electric field is of the form

$$\vec{E} = \frac{\rho}{4\pi\varepsilon_0} \int_0^H \int_0^{2\pi} \int_{R_1}^{R_2} \frac{\vec{r}}{|\vec{r}|^3} R\, dR\, d\varphi\, dz. \quad (\varepsilon_0 = \text{permittivity in vacuum})$$

My point of interest was how the systems will treat the triple integral.

We perform the integration and plot the 3rd coordinate of the strength versus z_0:

```
(%i6) fv:r_/sqrt(r_.r_)^3*R$
(%i7) E_:rho/(4*%pi*eps0)*
          integrate(
             integrate(
                integrate(trigsimp(fv),R,R1,R2),
             phi,0,2*%pi),
          z,0,H);
Is z0 zero or nonzero?p;
```

$$\text{(\%o7)}\quad \left[0, 0, \frac{\varrho\left(\sqrt{R2^2+H^2-2\,z0\,H+z0^2}-\sqrt{R2^2+z0^2}-\sqrt{R1^2+H^2-2\,z0\,H+z0^2}+\sqrt{R1^2+z0^2}\right)}{2\,\pi\,eps0\,H\,(R2^2-R1^2)}\right]$$



We treat the problem with TI-NspireCAS:



Sliders improve the presentation:



The Analyse-Tool supports finding the position for the maximum strength.

Example 3: Magnetic Field

A uniform rod of length l and mass m is placed on two parallel, horizontal rails. These rails are connected to a source of constant potential difference U and placed in a constant magnetic field of strength B. This magnetic field is perpendicular to the plane containing the rails. The coefficient of friction between the rod and the rails equals μ.

a) What is the velocity of the rod at time t?

b) Assuming that the rails are infinitely long, calculate the maximum velocity of the rod.



The motion of the rod is determined by a force $\vec{F}$ which is the resultant of two forces: the electromagnetic force $\overrightarrow{F_{el}}$ and the force of friction $\overrightarrow{F_f}$.

$$F = F_{el} - F_f \quad \text{where} \quad F_{el} = B\,I\,l \quad \text{and} \quad F_f = \mu G = \mu\, m\, g.$$

Motion of the rod causes a change in the flux of the magnetic field Φ and an electromotive force is induced

$$E = -\frac{d\Phi}{dt}, \quad \text{where} \quad \Phi = B\,S = B\,l\,x(t).$$

The current flowing in the circuit is given according to *Ohm*'s law by

$$I = \frac{U+E}{R}.$$

Now we can enter all above given relations

```
(%i7)    Φ(t):=B*l*x(t)$   E(t):=-diff(Φ(t),t)$
         I(t):=(U+E(t))/R$   F[el](t):=B*I(t)*l$
         F[f]:μ*G$   G:m*g$   F(t):=F[el](t)-F[f]$
```

and we apply *Newton*'s 2nd law (Force = Mass times Acceleration):

```
(%i8)    eq:m*diff(x(t),t,2)=F(t);
```

(eq) $$m\left(\frac{d^2}{d\,t^2}\,x(t)\right) = \frac{l\,B\left(U - l\left(\frac{d}{d\,t}\,x(t)\right)B\right)}{R} - G\,\mu$$

Two ways of calculating the requested velocity of the rod i.e. the unknown function $x'(t)$ are presented below.

Method 1: We separate the variables in equation `(eq)` and integrate wrt *t*.

```
(%i9)    eq/rhs(eq);
```

$$(\%o9)\quad \frac{m\left(\frac{d^2}{dt^2}x(t)\right)}{\frac{l\,B\left(U-l\left(\frac{d}{dt}x(t)\right)B\right)}{R}-G\,\mu}=1$$

```
(%i10)   integrate(%,t);
```

$$(\%o10)\quad -\frac{m\,R\log\left(\frac{l\,B\left(U-l\left(\frac{d}{dt}x(t)\right)B\right)}{R}-G\,\mu\right)}{l^2\,B^2}=t+\%c1$$

Then we solve the resulting equation for *x*'(*t*) …

```
(%i11)   gsoln:solve(%,diff(x(t),t));
```

$$(gsoln)\quad \left[\frac{d}{dt}x(t)=-\frac{G\,R\,\mu-l\,B\,U+R\,\%e^{-\frac{l^2\,t\,B^2}{m\,R}-\frac{\%c1\,l^2\,B^2}{m\,R}}}{l^2\,B^2}\right]$$

… and extract the right hand side which represents the velocity.

```
(%i12)   v(t):=rhs(gsoln[1])$
(%i13)   v(t);
```

$$(\%o13)\quad -\frac{G\,R\,\mu-l\,B\,U+R\,\%e^{-\frac{l^2\,t\,B^2}{m\,R}-\frac{\%c1\,l^2\,B^2}{m\,R}}}{l^2\,B^2}$$

When time tends to infinite we get maximal velocity. This fact can formally be confirmed by calculation of the limit of the velocity.

```
(%i14)   solve(diff(v(t),t)=0,t);
(%o14)   []
(%i16)   assume(l>0,m>0,R>0)$
         limit(v(t),t,inf);
Is B zero or nonzero?n;
```

$$(\%o16)\quad -\frac{G\,R\,\mu-l\,B\,U}{l^2\,B^2}$$

I'd like to give an additional graphic confirmation by entering numerical data and then plotting the velocity function:

```
(%i24)   G:5$ R:20$ µ:0.03$
         B:1$ U:2$ l:10$ m:1$ %c1:0$
(%i25)   ev(%o16);
(%o25)   0.17
```

Let's plot the velocity function:

```
(%i26)  ev(%o13);
(%o26)  -(20 %e^(-5 t) - 17.0)/100
(%i28)  plot2d(-(20*%e^(-5*t)-17)/100,[t,0,2]);
```



Method 2: We choose applying function `ode2`.

We enter equation `(eq)` from above directly and solve the differential equation.

```
(%i2)  de:m*(diff(v(t),t))=(l*B*(U-l*v(t)*B))/R-μ*G;
```

(de) $m\left(\frac{d}{dt}\,v(t)\right)=\frac{l\,B\,(U-l\,v(t)\,B)}{R}-G\,\mu$

```
(%i3)  gsoln:ode2(de,v(t),t)$
(%i4)  v(t):=rhs(gsoln)$
(%i5)  v(t);
```

(%o5) $\%e^{-\frac{l^2\,t\,B^2}{m\,R}}\left(\%c-\frac{\%e^{\frac{l^2\,t\,B^2}{m\,R}}\,(G\,R\,\mu-l\,B\,U)}{l^2\,B^2}\right)$

```
(%i6)  limit(%o5,t,inf);
Is l zero or nonzero?n;
  Is m positive or negative?p;
  Is B zero or nonzero?n;
  Is R positive or negative?p;
```

(%o6) $-\frac{G\,R\,\mu-l\,B\,U}{l^2\,B^2}$

It is charming to compare solving the ODE applying DERIVE and TI-NspireCAS as well.

In DERIVE we make use of built-in DSOLVE1(p,q,t,y,t0,v0) after rewriting the equation in the form $p(t,v) + q(t,v)\cdot v' = 0$, $v(t_0) = v_0$.

#1: CaseMode := Sensitive

#2: $v\_(t) := \left(\text{SOLUTIONS}\left(\text{DSOLVE1}\left(-\dfrac{l\cdot B\cdot(U - l\cdot v\cdot B)}{R} + \mu\cdot G,\ m,\ t,\ v,\ 0,\ 0\right),\ v\right)\right)_1$

#3: $v\_(t) := \dfrac{e^{-B^2\cdot l^2\cdot t/(R\cdot m)}\cdot(G\cdot R\cdot\mu - B\cdot U\cdot l)}{B^2\cdot l^2} + \dfrac{B\cdot U\cdot l - G\cdot R\cdot\mu}{B^2\cdot l^2}$

#4: [R :∊ Real (0, ∞), m :∊ Real (0, ∞)]

#5: $\lim_{t\to\infty} v\_(t) = \dfrac{B\cdot U\cdot l - G\cdot R\cdot\mu}{B^2\cdot l^2}$

TI-NspireCAS makes entering the equation easier but here it is not possible to find the limit for *t* tending to infinity. But we can see this by inspecting the exponential expression, of course!

$\text{deSolve}\left(m\cdot v'=\dfrac{l\cdot bb\cdot(uu-l\cdot v\cdot bb)}{rr}-\mu\cdot gg,t,v\right)$ $\quad v=c3\cdot e^{\frac{-l^2\cdot bb^2\cdot t}{m\cdot rr}}+\dfrac{l\cdot bb\cdot uu-gg\cdot rr\cdot\mu}{l^2\cdot bb^2}$

$v\_(t):=\text{right}\left(\text{deSolve}\left(m\cdot v'=\dfrac{l\cdot bb\cdot(uu-l\cdot v\cdot bb)}{rr}-\mu\cdot gg \text{ and } v(0)=0,t,v\right)\right)$ $\quad$ *Done*

$v\_(t)$ $\quad \dfrac{l\cdot bb\cdot uu-gg\cdot rr\cdot\mu}{l^2\cdot bb^2}-\dfrac{(l\cdot bb\cdot uu-gg\cdot rr\cdot\mu)\cdot e^{\frac{-l^2\cdot bb^2\cdot t}{m\cdot rr}}}{l^2\cdot bb^2}$

$\lim_{t\to\infty}(v\_(t))$ $\quad$ undef

$\lim_{t\to\infty}(v\_(t))|rr>0 \text{ and } m>0$ $\quad$ undef

This is the limit.

Example 4: Circuits - Resistances

Find the total resistance for the circuit presented in the figure.



Give the numerical result for the total resistance for $R_1 = 2\Omega$, $R_2 = 3\Omega$, $R_3 = 2\Omega$, $R_4 = 3\Omega$, $R_5 = 2\Omega$, $R_6 = 3\Omega$, $R_7 = 2\Omega$, $R_8 = 2\Omega$ and $R_9 = 2\Omega$.

We "simplify" the circuit by manipulating the resistors.

Note that Resistors R_5, R_6 and R_7 are connected in series. Therefore the above circuit can be replaced by a simpler form (see below).



where $R_A = R_5 + R_6 + R_7$.

Resistors R_A and R_4 are parallel connected. Hence the next simplification is shown:



where $\frac{1}{R_B} = \frac{1}{R_4} + \frac{1}{R_A}$.

As the resistors R_3, R_B and R_8 are series connected we then have:



with $R_C = R_3 + R_B + R_8$.

Resistors R_2 and R_C are parallel connected.



where $\frac{1}{R_D} = \frac{1}{R_2} + \frac{1}{R_C}$.

Finally total resistance R_T can calculated according the formula

$$R_T = R_1 + R_D + R_9.$$

Let us solve the above equation system.

```
(%i5)   eq1:R[A]=R[5]+R[6]+R[7];
        eq2:1/R[B]=1/R[4]+1/R[A];
        eq3:R[C]=R[3]+R[B]+R[8];
        eq4:1/R[D]=1/R[2]+1/R[C];
        eq5:R[T]=R[1]+R[D]+R[9];
```

(eq1) $R_A=R_7+R_6+R_5$

(eq2) $\frac{1}{R_B}=\frac{1}{R_A}+\frac{1}{R_4}$

(eq3) $R_C=R_B+R_8+R_3$

(eq4) $\frac{1}{R_D}=\frac{1}{R_C}+\frac{1}{R_2}$

(eq5) $R_T=R_D+R_9+R_1$

```
(%i7)   eqs:[eq1,eq2,eq3,eq4,eq5]$
        sol:solve(eliminate(eqs,[R[A],R[B],R[C],R[D]]),R[T])[1];
```

(sol) $$R_T=\Big(\big((R_7+R_6+R_5+R_4)R_8+(R_4+R_3+R_2)R_7+(R_4+R_3+R_2)R_6+(R_4+R_3+R_2)R_5+(R_3+R_2)R_4\big)R_9+\big((R_2+R_1)R_7+(R_2+R_1)R_6+(R_2+R_1)R_5+(R_2+R_1)R_4\big)R_8+\big((R_2+R_1)R_4+(R_2+R_1)R_3+R_1R_2\big)R_7+\big((R_2+R_1)R_4+(R_2+R_1)R_3+R_1R_2\big)R_6+\big((R_2+R_1)R_4+(R_2+R_1)R_3+R_1R_2\big)R_5+\big((R_2+R_1)R_3+R_1R_2\big)R_4\Big)\Big/\Big((R_7+R_6+R_5+R_4)R_8+(R_4+R_3+R_2)R_7+(R_4+R_3+R_2)R_6+(R_4+R_3+R_2)R_5+(R_3+R_2)R_4\Big)$$

Notice the `eliminate` in `%i17`. Only R_T is of interest for us.

The solution of the system is quite bulky!! Finally it is easy to carry out the calculation for the given numerical data and find the resulting total resistance:

```
(%i9)   data:[R[1]=2,R[2]=3,R[3]=2,R[4]=3,R[5]=2,
        R[6]=3,R[7]=2,R[8]=2,R[9]=2]$
        subst(data,sol);
```

(%o9) $R_T=\frac{547}{91}$

```
(%i10)  float(%);
```

(%o10) $R_T=6.010989010989011$

With DERIVE we can do in the same way, however we miss the "eliminate" command. So we receive the bulky solution containing all variables (see next page).

#6: $\left[RA = R5 + R6 + R7 \wedge RB = \frac{R4\cdot(R5 + R6 + R7)}{R4 + R5 + R6 + R7} \wedge RC = \frac{R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot}{R4 + R5 + R6 + R7}\right.$

$$\frac{R2\cdot(R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7))}{R2\cdot(R4 + R5 + R6 + R7) + R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7)} \wedge RT =$$

$$\frac{R1\cdot(R2\cdot(R4 + R5 + R6 + R7) + R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7)) + R2\cdot(}{R2\cdot(R4 + R5 + R6 + R}$$

$$\frac{+ R8 + R9) + (R5 + R6 + R7)\cdot(R8 + R9)) + R9\cdot(R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 +}{+ R7 + R8) + R8\cdot(R5 + R6 + R7)}$$

#7: $\left[RA = 7 \wedge RB = \frac{21}{10} \wedge RC = \frac{61}{10} \wedge RD = \frac{183}{91} \wedge RT = \frac{547}{91}\right]$

#8: $[[RA = 7 \wedge RB = 2.1 \wedge RC = 6.1 \wedge RD = 2.01098901 \wedge RT = 6.01098901]]$

We could also work stepwise without displaying the intermediate results:

#1: [CaseMode := Sensitive, InputMode := Word]

#2: $RA := R5 + R6 + R7$

#3: $RB := \left(\text{SOLUTIONS}\left(\frac{1}{RB} = \frac{1}{R4} + \frac{1}{RA}, RB\right)\right)_1$

#4: $RC := R3 + RB + R8$

#5: $RD := \left(\text{SOLUTIONS}\left(\frac{1}{RD} = \frac{1}{R2} + \frac{1}{RC}, RD\right)\right)_1$

#6: $RT = R1 + RD + R9$

#7: $RT = R1 -$

$$\frac{R3^2\cdot(R4 + R5 + R6 + R7)^2 + 2\cdot R3\cdot(R4 + R5 + R6 + R7)\cdot(R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5\sim}{(R4 + R5 + R6 + R7)\cdot(R2\cdot(R4 + R5 + R6 + R7) + \sim}$$

$$\frac{+ R6 + R7)) + R4^2\cdot(R5 + R6 + R7 + R8)^2 + 2\cdot R4\cdot R8\cdot(R5 + R6 + R7)\cdot(R5 + R6 + R7 + R8) \sim}{R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8) + R8\cdot(R5 + R6 + R7)) \sim}$$

$$\frac{+ R8^2\cdot(R5 + R6 + R7)^2}{} +$$

$$\frac{R3\cdot(R4 + R5 + R6 + R7) + R4\cdot(R5 + R6 + R7 + R8 + R9) + (R5 + R6 + R7)\cdot(R8 + R9)}{R4 + R5 + R6 + R7}$$

#8: $RT = \frac{547}{91}$

#9: $RT = 6.01098901$

Finally we substitute the given resistances.

If we enter the designations together with an equals-sign then we can see the intermediate results, too.

$$\#2: \quad \mathrm{RA} := \mathrm{R5} + \mathrm{R6} + \mathrm{R7}$$

$$\#3: \quad \left(\mathrm{RB} := \left(\mathrm{SOLUTIONS}\left(\frac{1}{\mathrm{RB}} = \frac{1}{\mathrm{R4}} + \frac{1}{\mathrm{RA}}, \mathrm{RB}\right)\right)_1\right) = \mathrm{RB} := \frac{\mathrm{R4}\cdot(\mathrm{R5} + \mathrm{R6} + \mathrm{R7})}{\mathrm{R4} + \mathrm{R5} + \mathrm{R6} + \mathrm{R7}}$$

$$\#4: \quad (\mathrm{RC} := \mathrm{R3} + \mathrm{RB} + \mathrm{R8}) = \mathrm{RC} := \mathrm{R3} - \frac{\mathrm{R5}^2 + 2\cdot\mathrm{R5}\cdot(\mathrm{R6} + \mathrm{R7}) + \mathrm{R6}^2 + 2\cdot\mathrm{R6}\cdot\mathrm{R7} + \mathrm{R7}^2}{\mathrm{R4} + \mathrm{R5} + \mathrm{R6} + \mathrm{R7}} + \mathrm{R5} + \mathrm{R6} + \mathrm{R7} + \mathrm{R8}$$

Below you can see the same procedure carried out with TI-NspireCAS.

$$\mathbf{ra} := r5+r6+r7 \;\blacktriangleright\; r5+r6+r7$$

$$\mathbf{rb} := \mathrm{right}\left(\mathrm{solve}\left(\frac{1}{\mathbf{rb}}=\frac{1}{r4}+\frac{1}{\mathbf{ra}},\mathbf{rb}\right)\right) \;\blacktriangleright\; \frac{r4\cdot(r5+r6+r7)}{r4+r5+r6+r7}$$

$$\mathbf{rc} := r3+\mathbf{rb}+r8 \;\blacktriangleright\; r3-\frac{(r5+r6+r7)^2}{r4+r5+r6+r7}+r5+r6+r7+r8 \;⚠$$

$$\mathbf{rd} := \mathrm{right}\left(\mathrm{solve}\left(\frac{1}{\mathbf{rd}}=\frac{1}{r2}+\frac{1}{\mathbf{rc}},\mathbf{rd}\right)\right)$$

$$\blacktriangleright\; \frac{r2\cdot\left(r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8\right)}{r2\cdot(r4+r5+r6+r7)+r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8} \;⚠$$

$$\mathbf{rt} := r1+\mathbf{rd}+r9$$

$$\blacktriangleright\; r1 - \frac{\left(r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8\right)^2}{\left(r2\cdot(r4+r5+r6+r7)+r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8)+(r5+r6+r7)\cdot r8\right)\cdot(r4+r5+r6+r7)} + \frac{r3\cdot(r4+r5+r6+r7)+r4\cdot(r5+r6+r7+r8+r9)+(r5+r6+r7)\cdot(r8+r9)}{r4+r5+r6+r7} \;⚠$$

$$\mathbf{rt}|r1=2 \text{ and } r2=3 \text{ and } r3=2 \text{ and } r4=3 \text{ and } r5=2 \text{ and } r6=3 \text{ and } r7=2 \text{ and } r8=2 \text{ and } r9=2$$

$$\blacktriangleright\; \mathbf{6.01099}$$

Example 5: Circuits – Resistances and Maximum Current

We have three identical cells with an electromotive force of E and an internal resistance of r. How should these cells be connected to each other to serve as battery, in order to give the maximum current across an external resistance of R?

I choose this example because it demonstrates the way to treat inequalities.

It is sufficient to consider and to compare the four configurations of connecting the cells to a circuit as illustrated in figures a – d.



The other possible ways of connecting the cells involve different polarization of the cells, and are thus less favourable.

Using *Kirchhoff*'s laws we get the following equations for the four circuits:

(a)

$$E = I_a\left(R + \frac{1}{\frac{1}{r}+\frac{1}{r}+\frac{1}{r}}\right) = I_a\left(R + \frac{r}{3}\right)$$

(b)

$$\begin{cases} 2E = I_1(r+r) + I_b R \\ E = (I_b - I_1)r + I_b R \end{cases}$$

(c)

$$\begin{cases} 2E = I_1 r + I_c r + I_c R \\ 2E = (I_c - I_1)r + I_c r + I_c R \end{cases}$$

(d)

$$3E = I_d\,(3r + R)$$

We enter the equations or the systems of equations and then solve them for the currents I_a to I_d .

To continue the comparison procedure of the currents we assign variable names to each of the solutions from above:

```
(%i14)  I[a]:(3*E)/(3*R+r)$
        I[b]:(4*E)/(3*R+2*r)$
        I[c]:(4*E)/(2*R+3*r)$
        I[d]:(3*E)/(R+3*r)$
```

We introduce a positive parameter ε defining the ratio of the resistances $\frac{R}{r}$ to make the comparison of the currents easier.

```
(%i15)  R:ε*r$
```

Circuit (a) is the most favourable if the following inequalities hold:

$$I_a > I_b, I_a > I_c, I_a > I_d \text{ i.e. } \frac{I_a}{I_b} > 1, \frac{I_a}{I_c} > 1, \frac{I_a}{I_d} > 1.$$

We solve the system of inequalities given above (loading a special package "`to_poly_solve`" first).

```
(%i16)  load(to_poly_solve)$
(%i17)  %solve(ev([I[a]/I[b]>1,I[a]/I[c]>1,I[a]/I[d]>1,ε>0]),ε);
to_poly_solve: to_poly_solver.mac is obsolete; I'm loading to_poly_solve.mac instead.
(%o17)  %union([0<ε, ε<2/3])
```

From `%o17` we can conclude that circuit (a) is the most favourable for a ratio of resistances in the range $0 < \varepsilon < \frac{2}{3}$ or $0 < \frac{R}{r} < \frac{2}{3}$ or $R > \frac{2r}{3}$.

The approach for the remaining cases is similar:

```
 same for circuit (b)
(%i18)  %solve(ev([I[b]/I[a]>1,I[b]/I[c]>1,I[b]/I[d]>1,ε>0]),ε);
(%o18)  %union([2/3<ε, ε<1])
 same for circuit (c)
(%i19)  %solve(ev([I[c]/I[a]>1,I[c]/I[b]>1,I[c]/I[d]>1,ε>0]),ε);
(%o19)  %union([1<ε, ε<3/2])
 and finally for circuit (d)
(%i20)  %solve(ev([I[d]/I[a]>1,I[d]/I[b]>1,I[d]/I[c]>1,ε>0]),ε);
(%o20)  %union([3/2<ε])
```

Circuit (b) is the best in the range $\frac{2}{3} < \frac{R}{r} < 1$ or $\frac{2r}{3} < R < r$,

Circuit (c) is the best for $1 < \frac{R}{r} < \frac{3}{2}$ or $r < R < \frac{3r}{2}$ and circuit (d) is the best for $\frac{R}{r} > \frac{3}{2}$ or $R > \frac{3r}{2}$.

As can easily be seen all intervals are open. We will check now how the circuits behave when the ratio of the resistances is at one of the boundaries of these intervals? What do you expect?

I will skip this here.

We might ask ourselves how DERIVE and/or TI-NspireCAS will perform solving the system of inequalities? We enter the definitions of the currents ($R = rr$) and make the try:

$$ia:=\frac{3\cdot e}{3\cdot rr+r}:ib:=\frac{4\cdot e}{3\cdot rr+2\cdot r}:ic:=\frac{4\cdot e}{2\cdot rr+3\cdot r}:id:=\frac{3\cdot e}{rr+3\cdot r} \qquad \frac{3\cdot e}{3\cdot r+rr}$$

$$rr:=\varepsilon\cdot r \qquad \varepsilon\cdot r$$

$$\text{solve}\left(\frac{ia}{ib}>1 \text{ and } \frac{ia}{ic}>1 \text{ and } \frac{ia}{id}>1,\varepsilon\right)|\varepsilon>0 \qquad 0<\varepsilon<\frac{2}{3}$$

$$\text{solve}\left(\frac{ib}{ia}>1 \text{ and } \frac{ib}{ic}>1 \text{ and } \frac{ib}{id}>1,\varepsilon\right) \qquad \frac{2}{3}<\varepsilon<1$$

$$\text{solve}\left(\frac{ic}{ia}>1 \text{ and } \frac{ic}{ib}>1 \text{ and } \frac{ic}{id}>1,\varepsilon\right) \qquad 1<\varepsilon<\frac{3}{2}$$

$$\text{solve}\left(\frac{id}{ia}>1 \text{ and } \frac{id}{ib}>1 \text{ and } \frac{id}{ic}>1,\varepsilon\right) \qquad \varepsilon<-3 \text{ or } \varepsilon>\frac{3}{2}$$

$$\text{solve}\left(\frac{ia}{ib}>1 \text{ and } \frac{ia}{ic}>1 \text{ and } \frac{ia}{id}>1,\varepsilon\right) \qquad \frac{-1}{3}<\varepsilon<\frac{2}{3}$$

As we can see there is no problem and we don't need any special package or library. The CAS-machine of TI-NspireCAS is – more or less – based on the DERIVE core, so we can be quite sure that DERIVE doesn't have any problems, too.

Entering the inequalities and output of the result are very clear.

We will enter the world of differential equations – this is where physics really starts …

Example 6: Circuits – R-L-Circuit

Given is a R-L circuit consisting of time dependent electromotive force of the form $V(t) = V_0 \cos(\omega t)$, a resistor R and an inductor L, connected in series.

a) Calculate the charge $Q(t)$ and the current $I(t)$ when $Q(0) = Q_0$ and $I(0) = 0$

b) Plot the graphs of $Q(t)$ and $I(t)$ for $Q_0 = V_0 = R = \omega = 1$, $L = 2$.

We have to solve the following differential equation: $L\,Q''(t) + R\,Q(t) = V_0 \cos(\omega t)$.

```
(%i1)   de:'L*diff(Q(t),t,2)+R*diff(Q(t),t)=V[0]*cos(ω*t);
```

(de) $$\left(\frac{d}{d\,t}\,Q(t)\right) R + \left(\frac{d^2}{d\,t^2}\,Q(t)\right) L = V_0 \cos(t\,\omega)$$

```
(%i2)   Qsoln:ode2(de,Q(t),t);
Is R zero or nonzero?n;
```

(Qsoln) $$Q(t) = \frac{V_0\,R\sin(t\,\omega) - V_0\,L\,\omega\cos(t\,\omega)}{L^2\,\omega^3 + R^2\,\omega} + \%k2\,\%e^{-\frac{t\,R}{L}} + \%k1$$

```
-->     Q(t):=(V[0]*R*sin(t*ω)-V[0]*L*ω*cos(t*ω))/
               (L^2*ω^3+R^2*ω)+%k2*%e^(-(t*R)/L)+%k1$
```

It remains to calculate the two constants of integration.

```
(%i5)   eq1:Q(0)=Q[0]$
        eq2:subst(t=0,diff(Q(t),t))=0$
(%i6)   solve([eq1,eq2],[%k1,%k2]);
```

(%o6) $$[\,[\,\%k1 = Q_0,\ \%k2 = \frac{V_0\,L}{L^2\,\omega^2 + R^2}\,]\,]$$

```
(%i7)   [%k1:Q[0],%k2:(V[0]*L)/(L^2*ω^2+R^2)]$
```

Now we can define charge $Q(t)$ and $I(t)$ as the derivative of the charge:

```
(%i8)   Q(t);
```

(%o8) $$\frac{V_0\,R\sin(t\,\omega) - V_0\,L\,\omega\cos(t\,\omega)}{L^2\,\omega^3 + R^2\,\omega} + \frac{V_0\,L\,\%e^{-\frac{t\,R}{L}}}{L^2\,\omega^2 + R^2} + Q_0$$

```
(%i9)   I(t):=diff(Q(t),t)$
(%i10)  I(t);
```

(%o10) $$\frac{V_0\,L\,\omega^2\sin(t\,\omega) + V_0\,R\,\omega\cos(t\,\omega)}{L^2\,\omega^3 + R^2\,\omega} - \frac{V_0\,R\,\%e^{-\frac{t\,R}{L}}}{L^2\,\omega^2 + R^2}$$

We calculate charge and current for the given data

```
(%i11)  example:subst([Q[0]=1,V[0]=1,R=1,L=2,ω=1],[Q(t),I(t)]);
```

(example) $$[\,\frac{\sin(t) - 2\cos(t)}{5} + \frac{2\,\%e^{-\frac{t}{2}}}{5} + 1,\ \frac{2\sin(t) + \cos(t)}{5} - \frac{\%e^{-\frac{t}{2}}}{5}\,]$$

Finally we prepare for plotting and then we plot the requested functions:

```
(%i13)  Q1(t):=(sin(t)-2*cos(t))/5+(2*%e^(-t/2))/5+1$
        I1(t):=(2*sin(t)+cos(t))/5-%e^(-t/2)/5$
-->     plot2d([Q1(t),I1(t)],[t,0,20],[y,-1,2],
            [legend,"Charge","Current"]);
```



Let's try Michel Beaudin's toolbox (see References). There is also a function provided for treating an R-L-circuit:

$kit\_ets\_mb\backslash cir\_rl(1,2,\cos(t),0)$ $\qquad \dfrac{-e^{\frac{-t}{2}}}{5}+\dfrac{\cos(t)}{5}+\dfrac{2\cdot \sin(t)}{5}$

$current(t):=\dfrac{-e^{\frac{-t}{2}}}{5}+\dfrac{\cos(t)}{5}+\dfrac{2\cdot \sin(t)}{5}$ $\qquad$ *Done*

$q(t):=\int current(t)\,dt+c$ $\qquad$ *Done*

$q(0)=1$ $\qquad c=1$

$charge(t):=q(t)|c=1$ $\qquad$ *Done*

$charge(t)$ $\qquad \dfrac{2\cdot e^{\frac{-t}{2}}}{5}-\dfrac{2\cdot \cos(t)}{5}+\dfrac{\sin(t)}{5}+1$

The plot looks the same as with wxMaxima.

Example 7: Electric Charge in Magnetic Field

Calculate the trajectory of a particle of mass m and electrical charge q moving in a constant magnetic field $\vec{B}$.
At time $t = 0$ position and velocity of the particle are: $\vec{r}_0 = (0,0,0)$ and $\vec{v}_0 = (v_{0x}, 0, v_{0z})$.

Let us assume the co-ordinate system oriented as given in the figure below.



Using such a co-ordinate system we can write the field components as follows:

$$B_x = 0,\ B_y = 0,\ B_z = B.$$

For the components of the initial velocity we have:

$$v_{0x} = v_0 \sin(\alpha), v_{0y} = 0,\ v_{0z} = v_0 \cos(\alpha).$$

We enter the definitions of position and velocity vector and of both field vectors.

```
(%i7)   r_(t):=[x(t),y(t),z(t)]$
        vx(t):='diff(x(t),t)$   vy(t):='diff(y(t),t)$
        vz(t):='diff(z(t),z)$
        v_(t):=[vx(t),vy(t),vz(t)]$
        E_:[0,0,0]$   B_:[0,0,B]$
```

We need the library `vector` for applying *Newton*'s equation $\ddot{\vec{r}} = \dfrac{q}{m}(\vec{v} \times \vec{B})$:

```
(%i7)   r_(t):=[x(t),y(t),z(t)]$
        vx(t):='diff(x(t),t)$   vy(t):='diff(y(t),t)$
        vz(t):='diff(z(t),z)$
        v_(t):=[vx(t),vy(t),vz(t)]$
        E_:[0,0,0]$   B_:[0,0,B]$
(%i8)   load(vect)$
(%i9)   de_:'diff(r_(t),t,2)=q*E_/m+q/m*express(v_(t)~B_);
```

$$\text{(de\_)}\quad \frac{d^2}{dt^2}[x(t), y(t), z(t)] = [\frac{B\,q\left(\frac{d}{dt}y(t)\right)}{m}, -\frac{B\,q\left(\frac{d}{dt}x(t)\right)}{m}, 0]$$

```
(%i10)  de_:subst(B=ω*m/q,de_);
```

$$\text{(de\_)}\quad \frac{d^2}{dt^2}[x(t), y(t), z(t)] = [\left(\frac{d}{dt}y(t)\right)\omega, -\left(\frac{d}{dt}x(t)\right)\omega, 0]$$

We substitute $\omega = \dfrac{B \cdot q}{m}$ to get more comfortable expressions and then extract differential Equations for all components.

We extract differential equations for all Cartesian components:

```
(%i11) dex2:'diff(x(t),t,2)=rhs(de_)[1];
```
(dex2) $$\frac{d^2}{d\,t^2}\,x(t)=\left(\frac{d}{d\,t}\,y(t)\right)\,\omega$$

```
(%i12) dey2:'diff(y(t),t,2)=rhs(de_)[2];
```
(dey2) $$\frac{d^2}{d\,t^2}\,y(t)=-\left(\frac{d}{d\,t}\,x(t)\right)\,\omega$$

```
(%i13) dez2:'diff(z(t),t,2)=rhs(de_)[3];
```
(dez2) $$\frac{d^2}{d\,t^2}\,z(t)=0$$

Method 1:

By integration: Solution of DE `dez2` is trivial:

$z(t=0)=0$ and $\dot{z}(t=0)=v0z$ give `%c2` $=0$, `%c1` $=v_0\cos(\alpha)\rightarrow z(t)=t\cdot v_0\cos(\alpha)$.

We proceed with calculating $x(t)$ and $y(t)$.

In the first step we integrate one of the two equations `dex2` or `dey2`, let's take the second one. Taking into account the initial condition $\dot{y}(0)=0$, it can easily be seen that constant `%c3` in `%o16` is equal zero. Therefore `%o16` can be written in simpler form %o17.

```
(%i14) integrate(dez2,t);
```
(%o14) $$\frac{d}{d\,t}\,z(t)=\%c1$$

```
(%i15) integrate(%,t);
(%o15) z(t)=%c1 t+%c2
(%i16) integrate(dey2,t);
```
(%o16) $$\frac{d}{d\,t}\,y(t)=\%c3-x(t)\,\omega$$

```
(%i17) 'diff(y(t),t)=-ω*x(t)$
```

In the next step we insert the above derivative into equation `dex2`.

```
(%i18) subst('diff(y(t),t,1)=-x(t)*ω,dex2);
```
(%o18) $$\frac{d^2}{d\,t^2}\,x(t)=-x(t)\,\omega^2$$

It is interesting that `ode2` does not return the correct solution, so I try `desolve` – and it works

```
(%i19) desolve('diff(x(t),t,2)=-omega^2*x(t),x(t));
Is ω zero or nonzero?n;
```
(%o19) $$x(t)=\frac{\sin(\omega\,t)\left(\left.\frac{d}{d\,t}\,x(t)\right|_{t=0}\right)}{\omega}+x(0)\cos(\omega\,t)$$

We can substitute $x'(0)$ and $x(0)$ by copy and past in `%o18` or we apply `subst`:

```
(%i20) subst([x(0)=0,at('diff(x(t),t,1),t=0)=v[0]*sin(α)],%);
```

(%o20) $$x(t)=\frac{v_0\sin(\omega t)\sin(\alpha)}{\omega}$$

Our next step is inserting the obtained function $x(t)$ in expression `%i17` followed by integrating to get function $y(t)$:

```
(%i21) subst(x(t)=(v[0]*sin(ω*t)*sin(α))/
              ω,'diff(y(t),t,1)=-ω*x(t));
```

(%o21) $$\frac{d}{d\,t}\,y(t)=-v_0\sin(\alpha)\sin(t\,\omega)$$

```
(%i22) integrate(%,t);
```

(%o22) $$y(t)=\frac{v_0\sin(\alpha)\cos(t\,\omega)}{\omega}+\%c4$$

In the final step we apply initial condition $y(0) = 0$ to find constant `%c4`. Then it's easy to get the simplified result for $y(t)$.

```
(%i23) solve(subst(t=0,(v[0]*cos(ω*t)*sin(α)*ω)/ω^2+%c4=0),%c4);
```

(%o23) $$[\,\%c4=-\frac{v_0\sin(\alpha)}{\omega}\,]$$

```
(%i24) y(t)=(v[0]*cos(ω*t)*sin(α)*ω)/ω^2-(v[0]*sin(α))/ω;
```

(%o24) $$y(t)=\frac{v_0\sin(\alpha)\cos(t\,\omega)}{\omega}-\frac{v_0\sin(\alpha)}{\omega}$$

```
(%i25) ratsimp(%);
```

(%o25) $$y(t)=\frac{v_0\sin(\alpha)\cos(t\,\omega)-v_0\sin(\alpha)}{\omega}$$

Method 2:

Using complex variables:

The system of equations `[dex2,dey2]` can be solved in an elegant way by introducing a new complex variable

$$\eta(t)=x(t)+i\cdot y(t)\ \text{ where } i \text{ denotes the imaginary unit.}$$

We start adding `dex2` and `%i·dey2`. Then we rewrite the resulting equation in form of a single equation for the complex function $\eta(t)$.

```
(%i26) B:ω*m/q$
(%i27) dex2+%i*dey2;
```

(%o27) $$\%i\left(\frac{d^2}{d\,t^2}\,y(t)\right)+\frac{d^2}{d\,t^2}\,x(t)=\left(\frac{d}{d\,t}\,y(t)\right)\omega-\%i\left(\frac{d}{d\,t}\,x(t)\right)\omega$$

```
(%i28) diff(η(t),t,2)=-%i*ω*diff(eta(t),t);
```

(%o28) $$\frac{d^2}{d\,t^2}\,\eta(t)=-\%i\left(\frac{d}{d\,t}\,\eta(t)\right)\omega$$

At first we perform the integration:

```
(%i29)  integrate(%,t);
```

$$\text{(\%o29)}\quad \frac{d}{d\,t}\,\eta(t) = \%c5 - \%i\,\eta(t)\,\omega$$

Taking into account the initial conditions

$$x(0)=0,\;\; y(0)=0,\;\; \dot{x}(0)=v_0\sin(\alpha),\;\; \dot{y}(0)=0$$

we get

$$\dot{\eta}(0)=\dot{x}(0)+i\,\dot{y}(0)=v_0\sin(\alpha)$$

and equation `%o29` can be written as

```
(%i30)  'diff(η(t),t)=v[0]*sin(α)-%i*(q*η(t)*B)/m;
```

$$\text{(\%o30)}\quad \frac{d}{d\,t}\,\eta(t) = v_0\,\sin(\alpha) - \%i\,\eta(t)\,\omega$$

We apply `ode2` - which can be applied for 1st order DEs, too – and define $\eta(t)$.

```
(%i31)  ode2(%,η(t),t);
```

$$\text{(\%o31)}\quad \eta(t) = \%e^{-\%i\,t\,\omega}\left(\%c - \frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega}}{\omega}\right)$$

```
(%i32)  η(t):=%e^(-%i*t*ω)*(%c-(%i*v[0]*sin(α)*%e^(%i*t*ω))/ω);
```

$$\text{(\%o32)}\quad \eta(t) := \%e^{(-\%i)\,t\,\omega}\left(\%c - \frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega}}{\omega}\right)$$

Considering the initial condition delivers constant `%c`.

```
(%i33)  solve(η(0)=0,%c);
```

$$\text{(\%o33)}\quad \left[\,\%c = \frac{\%i\,v_0\,\sin(\alpha)}{\omega}\,\right]$$

```
(%i34)  η(t):=%e^((-%i)*t*ω)*
            ((%i*v[0]*sin(α))/ω-(%i*v[0]*sin(α)*%e^(%i*t*ω))/ω);
```

$$\text{(\%o34)}\quad \eta(t) := \%e^{(-\%i)\,t\,\omega}\left(\frac{\%i\,v_0\,\sin(\alpha)}{\omega} - \frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega}}{\omega}\right)$$

```
(%i35)  ratsimp(%);
```

$$\text{(\%o35)}\quad \eta(t) := -\frac{\%i\,v_0\,\sin(\alpha)\,\%e^{\%i\,t\,\omega} - \%i\,v_0\,\sin(\alpha)}{\omega\,\%e^{\%i\,t\,\omega}}$$

```
(%i36)  η_(t):=expand(η(t));
(%o36)  η(t):=expand(η(t))
(%i37)  x(t)=realpart(η_(t));
```

$$\text{(\%o37)}\quad x(t) = \frac{v_0\,\sin(\alpha)\,\sin(t\,\omega)}{\omega}$$

```
(%i38)  y(t)=imagpart(η_(t));
```

$$\text{(\%o38)}\quad y(t) = \frac{v_0\,\sin(\alpha)\,\cos(t\,\omega)}{\omega} - \frac{v_0\,\sin(\alpha)}{\omega}$$

Same result as above!

It remains to extract real and imaginary part in order to obtain requested functions $x(t)$ and $y(t)$.

Method 3: Using `desolve`.

```
(%i39) gsoln:desolve([dex2,dey2,dez2],[x(t),y(t),z(t)])$
Is ω zero or nonzero?n;
(%i40) psoln:subst([x(0)=0,y(0)=0,z(0)=0,
                    at('diff(x(t),t),t=0)=v[0]*sin(α),
                    at('diff(y(t),t),t=0)=0,
                    at('diff(z(t),t),t=0)=v[0]*cos(α)],gsoln);
```

(psoln) $[x(t)=\frac{v_0\sin(\alpha)\sin(t\,\omega)}{\omega},\ y(t)=\frac{v_0\sin(\alpha)\cos(t\,\omega)}{\omega}-\frac{v_0\sin(\alpha)}{\omega},\ z(t)=v_0\,t\cos(\alpha)]$

```
(%i42) x(t):=rhs(psoln[1])$ x(t);
```

(%o42) $\frac{v_0\sin(\alpha)\sin(t\,\omega)}{\omega}$

```
(%i44) y(t):=expand(rhs(psoln[2]))$ y(t);
```

(%o44) $\frac{v_0\sin(\alpha)\cos(t\,\omega)}{\omega}-\frac{v_0\sin(\alpha)}{\omega}$

```
(%i46) z(t):=expand(rhs(psoln[3]))$ z(t);
```

(%o46) $v_0\,t\cos(\alpha)$

Comparing the results we fortunately can observe that they are the same.

Finally we will substitute back for $\omega=\frac{B\cdot q}{m}$.

```
(%i47) kill(B)$
(%i49) x_(t):=subst(ω=B*q/m,x(t))$ x_(t);
```

(%o49) $\frac{v_0\,m\sin\left(\frac{B\,q\,t}{m}\right)\sin(\alpha)}{B\,q}$

```
(%i51) y_(t):=subst(ω=B*q/m,y(t))$ y_(t);
```

(%o51) $\frac{v_0\,m\cos\left(\frac{B\,q\,t}{m}\right)\sin(\alpha)}{B\,q}-\frac{v_0\,m\sin(\alpha)}{B\,q}$

Now having done all the work we would like to see the trajectory of the particle for one data set:

```
(%i52) subst([v[0]=5,α=%pi/3,B=3,m=2,q=1],[x_(t),y_(t),z(t)]);
```

(%o52) $[\frac{5\sin\left(\frac{3t}{2}\right)}{\sqrt{3}},\ \frac{5\cos\left(\frac{3t}{2}\right)}{\sqrt{3}}-\frac{5}{\sqrt{3}},\ \frac{5t}{2}]$

```
(%i53) spiral:parametric((5*sin((3*t)/2))/sqrt(3),
                         (5*cos((3*t)/2))/sqrt(3)-5/sqrt(3),
                         (5*t)/2,t,0,3*%pi);
```

(spiral) $\text{parametric}\left(\frac{5\sin\left(\frac{3t}{2}\right)}{\sqrt{3}},\ \frac{5\cos\left(\frac{3t}{2}\right)}{\sqrt{3}}-\frac{5}{\sqrt{3}},\ \frac{5t}{2},\ t,\ 0,\ 3\pi\right)$



Graphs of some other trajectories:









Example 8: Challenge provided by Michel Beaudin

(This ione of Michel's assessment problems given at ETS Montreal)

Recall: the ODE for a mass-spring problem is

$$my'' + by' + ky = f(t),\ \ y(0) = y_0, \quad y'(0) = v_0$$

where $y(t)$ denotes the position of the object at time t, m is the mass of the object, b is the damping constant, k is the spring constant and $f(t)$ is the external force (could be 0) and where the initial position and initial velociy are respectiveley y_0 and v_0.

Problem 1 : consider the (undamped) mass-spring problem with 2 impulses acting as external force :

$$y'' + 4y = 50\delta(t-\pi) - 100\,\delta(t-2\pi), \quad y(0) = 10, \ \ y'(0) = 5\,.$$

a) Solve the ODE and plot the graph of the position in the window

$$0 < t < 10\pi, \ \ -30 < y < 30.$$

b) For $t > 2\pi$, show that the solution can be written as $A\cos(\omega t + \varphi)$.

I must admit, it is a shame but I never had to cope with Laplace transforms, Dirac Delta functions, … So I was busy informing myself. It was a steep learning curve but finally I was successful. See first how I did with wxMaxima. Then I will present Michel's solution.

I perform the Laplace transformation applied on the given differential equation:

```
(%i1) laplace(diff(y(t),t,2)+4*y(t)=
              50*delta(t-%pi)-100*delta(t-2*%pi),t,s);
```

(%o1) $-\left.\frac{d}{d\,t}\,y(t)\right|_{t=0} + s^2\,\mathrm{laplace}(y(t),t,s) + 4\,\mathrm{laplace}(y(t),t,s) - y(0)\,s = 50\,\%e^{-\pi s} - 100\,\%e^{-2\pi s}$

```
(%i2) eq_L:subst([-(at('diff(y(t),t,1),t=0)=5),
              y(0)=10,laplace(y(t),t,s)=Y],%o1);
```

(eq_L) $Y\,s^2 - 10\,s + 4\,Y - 5 = 50\,\%e^{-\pi s} - 100\,\%e^{-2\pi s}$

```
(%i3) solve(eq_L,Y);
```

(%o3) $\left[Y = \frac{\%e^{-2\pi s}\left((10\,s+5)\,\%e^{2\pi s} + 50\,\%e^{\pi s} - 100\right)}{s^2+4}\right]$

I separate the rational expression and apply the inverse Laplace transform, giving %o7.

```
(%i4) nom:expand(%e^(-2*%pi*s)*((10*s+5)*%e^(2*%pi*s)
              +50*%e^(%pi*s)-100));
```

(nom) $50\,\%e^{-\pi s} - 100\,\%e^{-2\pi s} + 10\,s + 5$

```
(%i5) ilt((10*s+5)/(s^2+4),s,t);
```

(%o5) $\frac{5\sin(2\,t)}{2} + 10\cos(2\,t)$

I treat the remaining two fractions according to the rules giving products including the step-function.

```
(%i6)   unit_step(t-%pi)*ilt(50/(s^2+4),s,t-%pi)+
           unit_step(t-2*%pi)*ilt(-100/(s^2+4),s,t-2*%pi);
(%o6)   25 unit_step(t-π) sin(2 (t-π))-50 unit_step(t-2 π) sin(2 (t-2 π))
```

Finally I tried to automate this process, transforming back expressions generated by the δ-function in the given DE.

```
(%i7)   invl(a_,f,s,t):=trigsimp(unit_step(t-a_)*ilt(f,s,t-a_))$
(%i8)   invl(%pi,50/(s^2+4),s,t)+invl(2*%pi,-100/(s^2+4),s,t);
(%o8)   25 sin(2 t) unit_step(t-π)-50 sin(2 t) unit_step(t-2 π)
(%i9)   ms(t):=(5*sin(2*t))/2+10*cos(2*t)+
             25*unit_step(t-%pi)*sin(2*(t-%pi))-
             50*unit_step(t-2*%pi)*sin(2*(t-2*%pi))$
(%i10)  plot2d(ms(t),[t,0,10*%pi]);
```

This is the plot:



Michel's TI-NspireCAS solution is given below.

The students are permitted to apply functions provided in the ETS-library like ressort().

Problem 1: the ODE is $y'' + 4y = 50\cdot\delta(t-\pi) - 100\cdot\delta(t-2\pi)$, $y(0) = 10$, $y'(0) = 5$.

The solution is:

kit_ets_mb\ressort$(1,0,4,50\cdot\delta(t-\pi)-100\cdot\delta(t-2\cdot\pi),10,5)$

$\blacktriangleright\ 10\cdot\cos(2\cdot t)\cdot u(t)+\sin(2\cdot t)\cdot\left(25\cdot u(t-\pi)-50\cdot u(t-2\cdot\pi)+\frac{5\cdot u(t)}{2}\right)$

Now, in order to plot this, we need to "inform" Nspire about $u(t)$. It is the Heaviside (unit−step) function that is programmed into the ets_specfunc library BUT not taken in charge by Nspire CAS alone… But the signum function is implemented into Nspire CAS. So let's define

$\mathbf{uu}(t):=\frac{1+\text{sign}(t)}{2}$ ▸ *Done*

And let's change "u" for "uu" in the answer, change "t" for "x" and plot the graph, using the indicated window.

$\mathbf{f1}(x):=10\cdot\cos(2\cdot x)\cdot\mathbf{uu}(x)+\sin(2\cdot x)\cdot\left(25\cdot\mathbf{uu}(x-\pi)-50\cdot\mathbf{uu}(x-2\cdot\pi)+\frac{5\cdot\mathbf{uu}(x)}{2}\right)$ ▸ *Done*

See page 2 where the effect of each impulse is clear on the graph.

Now, when $t > 2\pi$, $\mathbf{f1}(t)$ will simplify:

In his solutions is the oscillation function expressed as piecewise defined function. Can we achieve this with Maxima, too?



La fonction « ressort » donne ce résultat immédiatement :

$kit\_ets\_mb\backslash ressort(1,0,4,50\cdot\delta(t-\pi)-100\cdot\delta(t-2\cdot\pi),10,5)$

$$10\cdot\cos(2\cdot t)\cdot u(t)+\sin(2\cdot t)\cdot\left(25\cdot u(t-\pi)-50\cdot u(t-2\cdot\pi)+\frac{5\cdot u(t)}{2}\right)$$

Remarquez qu'on peut remettre cette réponse en « morceaux » :

$$y(t)=\begin{cases}10\cos(2t)+\dfrac{5}{2}\sin(2t), & 0\le t<\pi\\ 10\cos(2t)+\dfrac{55}{2}\sin(2t), & \pi\le t<2\pi\\ 10\cos(2t)-\dfrac{45}{2}\sin(2t), & t\ge 2\pi\end{cases}$$

The function which converts the Heaviside functions into a piecewise defined function was provided by Frederick Henri (ETS Montreal) and is also part of the TI-Nspire library.

Back to the above question:

Yes we can express the answer as a piecewise defined function with wxMaxima:

(%i12) `assume(t<%pi)$ ms(t);`

(%o12) $\dfrac{5\sin(2t)}{2}+10\cos(2t)$

(%i15) `forget(t<%pi)$ assume(t>%pi and t<2*%pi)$`
`trigsimp(ms(t));`

(%o15) $\dfrac{55\sin(2t)+20\cos(2t)}{2}$

(%i18) `forget(t>%pi and t<2*%pi)$ assume(t>2*%pi)$`
`trigsimp(ms(t));`

(%o18) $-\dfrac{45\sin(2t)-20\cos(2t)}{2}$

(%i19) `forget(t>2*%pi)$`

Notice the nice "forget"-function!

Now we can build the solution function in the requested form and then plot its graph again.

```
(%i22)  f1(t):=(5*sin(2*t))/2+10*cos(2*t)$
        f2(t):=(55*sin(2*t)+20*cos(2*t))/2$
        f3(t):=-(45*sin(2*t)-20*cos(2*t))/2$
(%i23)  ms2(t):=if t<%pi then f1(t) else
                 if t<2*%pi then f2(t) else f3(t)$
(%i25)  plot2d(ms2(t),[t,0,10*%pi]);
```



Question b) needs some competence in trig manipulations supported by the CAS:

$\mathbf{f1}(t)|t>2\cdot\pi \;\blacktriangleright\; 10\cdot\cos(2\cdot t)-\frac{45\cdot\sin(2\cdot t)}{2}$

And

$\text{tCollect}\left(10\cdot\cos(2\cdot t)-\frac{45\cdot\sin(2\cdot t)}{2}\right) \;\blacktriangleright\; \frac{-5\cdot\sqrt{97}\cdot\sin\left(2\cdot t-\tan^{-1}\left(\frac{4}{9}\right)\right)}{2}$

this is the same as $\frac{5\cdot\sqrt{97}\cdot\sin\left(2\cdot t-\tan^{-1}\left(\frac{4}{9}\right)+\pi\right)}{2} \;\blacktriangleright\; \frac{-5\cdot\sqrt{97}\cdot\sin\left(2\cdot t-\tan^{-1}\left(\frac{4}{9}\right)\right)}{2}$

And this is the same as

$\frac{5\cdot\sqrt{97}\cdot\cos\left(2\cdot t-\tan^{-1}\left(\frac{4}{9}\right)+\pi-\frac{\pi}{2}\right)}{2} \;\blacktriangleright\; \frac{-5\cdot\sqrt{97}\cdot\sin\left(2\cdot t-\tan^{-1}\left(\frac{4}{9}\right)\right)}{2}$

That is: $\frac{5\cdot\sqrt{97}\cdot\cos\left(2\cdot t-\tan^{-1}\left(\frac{4}{9}\right)+\pi-\frac{\pi}{2}\right)}{2} \;\blacktriangleright\; 24.6221\cdot\cos(2.\cdot t+1.15257)$

As this conference series was started in 1992 as a DERIVE conference I'd like to finish with the DERIVE treatment of this challenge.

The Dirac-δ-function has not been implemented in DERIVE. 2006 – after ending the "official life" of DERIVE Albert Rich wrote:

The Dirac delta function can be defined as the derivative of the step function. But because of the nature of this discontinuous function, it would have to be primitively defined in Derive in order to have the desired properties.

Michel Beaudin offers a trick to implement the δ-function as a limit of the CHI-function. See how it works and how to solve the differential equation – due to the fact, that DERIVE is able to perform integration of the CHI-function (which is internally based on the SIGN-function).

$$\delta(a) := \frac{1}{b}\cdot\chi(a,\ t,\ a + b)$$

```
DSOLVE2_IV(0, 4, 50·δ(π) - 100·δ(2·π), t, 0, 10, 5)
```

$$\lim_{b\to +0} \text{DSOLVE2\_IV}(0,\ 4,\ 50\cdot\delta(\pi) - 100\cdot\delta(2\cdot\pi),\ t,\ 0,\ 10,\ 5)$$

$$- 25\cdot\text{SIGN}(t - 2\cdot\pi)\cdot\text{SIN}(2\cdot t) + \frac{25\cdot\text{SIGN}(t - \pi)\cdot\text{SIN}(2\cdot t)}{2} + 10\cdot\text{COS}(2\cdot t) - 10\cdot\text{SIN}(2\cdot t)$$

The solution is the red graph. The slider for *b* demonstrates the property of the δ-function.



My Conclusions

- **Computer algebra systems significantly enhance the use of computers in teaching physics, far beyond simple 'number crunching'.**
- **Instead of spending time on algebraic manipulations with pen and paper, students (and practitioners alike) can tackle more challenging problems.**
- **With constant improvements and new developments in the area of CAS', their potential and impact on problem solving and teaching - not only in physics - will increase too.**



Computer Algebra
for Physics Examples

Electrostatics, Magnetism, Circuits and
Mechanics of Charged Particles
Part 1

Vector- and Scalar Fields
Electric Field

Leon Magiera and Josef Böhm

Computer Algebra
for Physics Examples

Electrostatics, Magnetism, Circuits and
Mechanics of Charged Particles
Part 2

Magnetic Field

Leon Magiera and Josef Böhm

Computer Algebra
for Physics Examples

Electrostatics, Magnetism, Circuits and
Mechanics of Charged Particles
Part 3

Circuits

Leon Magiera and Josef Böhm

Computer Algebra
for Physics Examples

Electrostatics, Magnetism, Circuits and
Mechanics of Charged Particles
Part 4

Mechanics of Charged Particles

Leon Magiera and Josef Böhm

You are invited to download the papers from http://rfdz.ph-noe.ac.at/acdca/materialien.html

Many thanks to Leon Magiera and Michel Beaudin for their patience and cooperation during preparing this lecture.

Leon Magiera

Thanks for your attention.

All files are available on request.

References:

https://sourceforge.net/projects/wxmaxima/
http://www.t3europe.eu/resources/engineering-mathematics/laplace-transforms/
https://cours.etsmtl.ca/SEG/mbeaudin/
http://seg-apps.etsmtl.ca/nspire/
http://www.austromath.at/dug/